Motorola Semiconductor Application Note

AN1218

HC05 to HC08 Optimization

By Mark Glenewinkel CSIC Applications Austin, Texas

Introduction

Motorola's HC05 Family of microcontrollers contains the world's most popular 8-bit microcontroller units (MCUs). In keeping pace with technology and the changing needs of the customer, Motorola has designed the HC08 Family of MCUs. The HC08 Family CPU is a performance extension to the HC05 Family of low cost MCUs. This application note will describe the differences and advantages of the HC08 Family CPU: the CPU08.

CPU08 is fully opcode and object code compatible with the HC05 CPU. Any HC05 code will execute directly on the HC08 without instruction set differences. As this application note will show, there are many improvements to the speed and capability in the CPU08.

CPU08 is a faster processor. The basic execution speed of the CPU08 has been increased with advanced high performance CMOS technology. Execution cycles of most instructions have been improved with an advanced computer architecture.

CPU08 has more programming capability. It has more addressing modes, better math support, and much improved data manipulation, accessing, and moving capabilities. Looping and branching instructions have also been optimized. This application note will help inform and educate the reader concerning the differences between the HC05 and HC08 CPUs. Detailed examples illustrating the added features found with the CPU08 are given to help optimize software design with the CPU08.

Scope of this Application Note

This note assumes the reader has a background in MCU software and hardware design and is also familiar with the HC05. It was written for the engineering manager and the design engineer. As a reference, the application note overviews the basic differences between the two CPUs so that one can fit the right CPU for a specific application. As a tutorial, the application note gives the designer the means to understand and utilize the HC08 enhancements. Software is given to illustrate and compare the performance of the CPUs.

HC08 Features

The following is a list of major features of the HC08 CPU (CPU08) that differentiate it from the HC05 CPU (CPU05).

- Fully upward object code compatible with the MC6805, MC146805, and the MC68HC05 Family
- 64 KByte program/data memory space
- Enhanced HC05 programming model
- 8 MHz CPU bus frequency
- 16 addressing modes, 5 more than the HC05
- Expandable internal bus definition for addressing range extension beyond 64 KBytes
- 16-bit index register with manipulation instructions
- 16-bit stack pointer with manipulation instructions
- Memory to memory data moves without using the accumulator

- Fast 8-bit multiply and integer/fractional divide instructions
- Binary coded decimal (BCD) instruction enhancements
- Internal bus flexibility to accommodate CPU enhancing peripherals such as a DMA controller
- Fully static low voltage/low power design

CPU05/CPU08 Programmer's Model Comparison

The CPU05 and the CPU08 programmer's model differences are illustrated in **Figure 1**.

- H Index Register The index register of the CPU08 has been extended to 16 bits, allowing the user to index or address a 64 KByte memory space without any offset. The upper byte of the index register is called the H index register. The concatenated 16-bit register is called the H:X register. Source code written for CPU05 will not affect the H register and it will remain in its reset state of \$00. There are seven new instructions that allow the user to manipulate the H:X index register. These instructions are covered in detail later.
- Stack PointerThe stack pointer (SP) has been extended from its 6-bit CPU05 version
to a full 16-bit SP on the CPU08. SPH:SPL refers to the 16-bit stack
pointer by naming the high byte, SPH, and the low byte, SPL. To
maintain HC05 compatibility, the reset state is \$00FF.

New instructions and new addressing modes greatly increase the utility of the CPU08 stack pointer over the CPU05 stack pointer. Nine new CPU08 instructions allow the user to easily manipulate the SP and the stack.

CPU08 also has relative addressing modes that allow the SP to be used as an index register to access temporary variables on the stack. These addressing modes and new instructions are discussed later in this application note.




Figure 1. CPU05 and CPU08 Comparison

Program CounterThe CPU08 program counter (PC) has been expanded to 16 bits which
allows the CPU08 to address 64 KBytes of memory. Not all HC05
devices have a 16-bit program counter.

New AddressingCPU08 has 16 addressing modes, 8 more than the HC05. Table 1 listsModes,these addressing modes and the CPUs that use them. A briefComparisondiscussion of these modes is given below.

Addressing Mode	HC05	HC08
Inherent	X	X
Immediate	X	Х
Direct	X	Х
Extended	X	X
Indexed, no offset	X	Х
Indexed, 8-bit offset	X	Х
Indexed, 16-bit offset	X	Х
Relative	X	Х
Stack Pointer, 8-bit offset		Х
Stack Pointer, 16-bit offset		Х
Memory to memory (4 modes)		Х
Indexed w/post increment		X
Indexed, 8-bit offset, w/post increment		X

Table 1. Addressing Mode Comparison Table

HC05 and HC08 Addressing Modes

Inherent instructions such as reset stack pointer (RSP) and multiply (MUL) have no operand. Inherent instructions require no memory address and are one byte long.

Immediate instructions contain a value that is used in an operation with the index register or accumulator. Immediate instructions require no memory address and are two bytes long. The operand is found in the byte immediately following the opcode.

Direct instructions can access any of the first 256 memory addresses with only two bytes. The first byte contains the opcode followed by the low byte of the operand address. The CPU automatically uses \$00 for the high byte of the operand address. Most direct instructions are two bytes long.

Application Note

Extended instructions can access any address in the memory map. Extended instructions are three bytes long and contain the opcode and the two-byte operand address.

Indexed instructions with no offset are one-byte instructions that utilize the index register of the CPU. CPU08 also uses the H:X register containing the high byte of the address operand.

Indexed, 8-bit offset instructions are two-byte instructions that utilize the index register of the CPU to access data at any location in memory. The 8-bit unsigned offset following the opcode is added to the 16-bit unsigned index register (H:X). The sum is the address used to access data.

Indexed, 16-bit offset instructions are like the 8-bit offset instructions except that they are three bytes long and add a 16-bit unsigned number to the 16-bit index register (H:X).

Relative addressing is only used for branch instructions. If the branching condition is true, the CPU finds the branch destination by adding the offset operand to the PC counter. The offset is a two's complement byte that gives a branching range of -128 to +127 bytes. This instruction is two bytes long.

New HC08Stack pointer, 8-bit offset instructions operate like indexed, 8-bit offsetAddressing Modesinstructions except that they add the offset to the 16-bit SP. This modeis available only on the CPU08. If interrupts are disabled, thisaddressing mode allows the SP to be used as a second index register.This instruction is three bytes long.

Stack pointer, 16-bit offset instructions are only available on the CPU08. They are like the stack pointer, 8-bit offset instructions except that they add a 16-bit value to the SP. This instruction is four bytes long. Memory to memory instructions utilize four different modes available only to the CPU08.

- 1. The move, immediate to direct, is a three-byte mode generally used to initialize RAM and register values in page 0 of the memory map. The operand in the second byte is immediately stored to the direct page location found in the third byte.
- 2. The move, direct to direct, is a three-byte instruction. The operand following the opcode is the direct page location that is stored to the second operand direct page location.
- The move, indexed to direct, post increment, is a two-byte instruction. The operand addressed by the 16-bit index register (H:X) is stored to direct page location address by the byte following the opcode. The index register is then incremented.
- 4. The move, direct to indexed, post increment, is a two-byte instruction. The operand in the direct page location addressed by the byte following the opcode is stored in the location addressed by the 16-bit index register (H:X). The index register is then incremented.

In the CPU08, four instructions address operands with the index register and then increment the index register afterwards. This is called indexed with post increment mode. These instructions include CBEQ indexed, CBEQ indexed with offset, MOV IX+Dir, and MOV Dirlx+.

 Table 2 gives examples to illustrate these different addressing modes.

Addressing Mode	Example		
Inherent	RSP		
Immediate	LDA	#\$FF	
Direct	LDA	\$50	
Extended	LDA	\$1000	
Indexed, no offset	LDA	,Х	
Indexed, 8-bit offset	LDA	\$50,X	
Indexed, 16-bit offset	LDA	\$0150,X	
Relative	BRA	\$20	
Stack Pointer, 8-bit offset*	LDA	\$50,SP	
Stack Pointer, 16-bit offset*	LDA	\$0150,SP	
Memory to memory ImmDir* DirDir* Ix+Dir* Dirlx+*	MOV MOV MOV MOV	#\$30,\$80 \$80,\$90 X+,\$90 \$80,X+	
Indexed w/post increment*	CBEQ	X+,LOOP	
Indexed, 8-bit offset, w/post increment*	CBEQ	\$20,X+,LOOP	

* New CPU08 addressing modes

Condition Code Register with Overflow Bit V

A summary of the condition code register (CCR) is given below. Unless otherwise stated, all bits correspond to both CPUs.

Overflow Bit V

This bit is set when a two's-complement overflow has occurred as the result of an operation. The V bit has been added to the CPU08 condition code register to support two's-complement arithmetic.

Half-Carry Bit H

The half-carry bit is set when a carry has occurred between bits 3 and 4 of the accumulator because of the last ADD or ADC operation. This bit is required for BCD operations.

Interrupt Mask Bit I

All timer and external interrupts are disabled when this bit is set. Interrupts are enabled when the bit is cleared. This bit is automatically set after any CPU reset.

Negative Bit N

This bit is set after any arithmetic, logical, or data manipulation operation was negative. In other words, bit 7 of the result of the operation was a logical one.

Zero Bit Z

The zero bit is set after any arithmetic, logical, or data manipulation operation was zero.

Carry/Borrow Bit C

The carry/borrow bit is set when a carry out of bit 7 of the accumulator occurred during the last arithmetic, logical, or data manipulation operation. The bit is also set or cleared during bit test and branch instructions and shifts and rotates.

Description of the Clock

In the CPU08, the CPU clock rate is twice that of the address/data bus rates. The internal CPU08 clock rate is 16 MHz for an 8 MHz HC08. To maintain a 50% duty cycle CPU clock, the oscillator clock, OSC CLK, must run twice the rate of the CPU clock. Therefore a 32 MHz OSC clock is needed to drive an 8 MHz HC08.

The flagship member of the CPU08 family has a phase locked loop (PLL) synthesizer to generate the 32 MHz signal. It is derived from a suggested crystal frequency of 4.9152 MHz.

Address/Data Rate	=	Ζ	=	8 MHz
CPU Clock Rate	=	2Z	=	16 MHz
OSC Clock Rate	=	4Z	=	32 MHz

Index Registers

	CPU08 has the additional H index register which is the high byte extension to the X index register. Together, the two index registers formulate the concatenated 16-bit H:X index register. Five new instructions are introduced on the CPU08 to allow manipulation of the H:X index register. Source code written for the HC05 will not effect the H register and it will remain in its reset state of \$00. The TSX and the TXS instructions also utilize the H:X index register. These instructions are covered in more detail in the stack pointer section.		
Five New Indexing Instructions, Detail	The new CPU08 instructions that affect the index registers are listed below. Examples for these instructions are given in Appendix A — New CPU08 Indexing Instruction Examples.		
	AIX	Add Immedia	ate to Index Register
		Operation:	$X \gets (H{:}X) + (M)$
		Description:	AIX adds an immediate value to the 16-bit index register formed by the concatenation of the H and X registers. The immediate operand is an 8-bit two's complement signed offset. Prior to addition to H:X, the offset is sign extended to 16 bits.
	CLRH	Clear Index	High
		Operation:	H ← \$00
		Description:	The contents of H are replaced with zeros.
	СРНХ	 Compare 16-bit Index Register Operation: (H:X) – (M:M+1) 	
		Description:	CPHX compares the 16-bit index register H:X with the 16-bit value in memory and sets the condition code register accordingly.

LDHX	Load 16-bit I	ndex Register
	Operation:	H:X ← (M:M+1)
	Description:	Loads the contents of the specified memory location into the 16-bit index register H:X. The condition codes are set according to the data.
STHX	Store 16-bit	Index Register
	Operation:	$(M:M+1) \leftarrow (H:X)$

Description: Stores the 16-bit index register H:X to the specified memory location. The condition codes are set according to the data.

Software
Techniques Using
IndexedThe CPU08 index register has some distinct advantages over the
CPU05 index register. Even though the CPU05 has 16-bit index offset,
the 8-bit index register restricts indexing to a maximum of 256 bytes.Addressing, TablesCPU08 with its H register extension allows full 16-bit index addressing
equaling 65,536 bytes of memory access. Proper 16-bit pointers allow
efficient compiling of C code and other higher level languages.
Maximum table lengths in the CPU08 which can be accessed in a single
instruction are therefore 64 KByte. An optional address extension
module can extend the data space beyond 64 KBytes, but the maximum
offset remains 64 KBytes. Index addressing modes include 8- and 16-
bit offsets.

Many programmers like to use calculated addressing. CPU08 has a new instruction, AIX, that allows the addition of a two's complement number. Table access is easier and more flexible.

The H:X index register can also be used as an auxiliary 16-bit accumulator. Sixteen-bit data comparisons are easier with the CPHX instruction.

The following section illustrates the advantage of using a 16-bit index register.

Code Example We will now illustrate the added benefit of the CPU08 16-bit index register. The index will be used to address a 512 byte table. In the CPU05, the table must be broken up into sections of memory consisting of 256 bytes per section. Our table has 512 bytes, so we will be using two sections, section 0 and section 1, for the CPU05. The address to look up on the table will be found in RAM. Notice that the CPU05 code is longer. If your table was larger, you would require more sections of memory to handle your table. A subroutine might be written to make the job more modular. In the HC08 example, the 512 byte table can be handled directly. A comparison between CPU05 and CPU08 code is shown in Appendix B — CPU05 and CPU08 512-Byte Table Indexing Code.

Stack Pointer

CPU08 has a full 16-bit stack pointer. To maintain compatibility with the CPU05, it is initialized to \$00FF out of reset.

Stack manipulation is from high to low memory. The SP is decremented each time data is pushed on the stack and incremented each time data is pulled from the stack. The SP points to the next available stack address rather than the latest stack entry address.

Nine new instructions have been added for the user to manipulate the stack. These instructions allow the direct push and pull of any register to the stack. The SP can be changed with a transfer of the H:X register to the SP or the SP can be augmented by the add immediate instruction.

Stack manipulation can be a very powerful programming technique. With the CPU08, the assembly programmer can pass parameters and store local or temporary variables when using subroutines and/or interrupts.

New addressing modes were added to address these variables on the stack. Using the stack pointer as an index register with 8- or 16-bit offsets, the user may access variables on the stack. These instructions greatly cut cycle count by not having to load/store the variable. RAM

requirements are also reduced. Significant C code efficiency can be gained when utilizing these new stack pointer addressing modes.

If interrupts are disabled, the stack pointer can be used as a second 16-bit index register with 8- or 16-bit offsets.

Nine New Stack Manipulation Instructions, Detail All the new CPU08 instructions that affect the stack pointer are listed below. Examples for these instructions are given in **Appendix C** — New CPU08 Stack Pointer Instructions.

AIS	Add Immedia	ate to Stack Pointer
	Operation:	$SP \gets (SP) + (M)$
	Description:	Adds the immediate operand to the stack pointer SP. The immediate value is an 8-bit two's complement signed operand. Prior to addition to the SP, the operand is sign extended to 16 bits. This instruction can be used to create and remove a stack frame buffer which is used to store temporary variables.
PSHA	Push Accum	ulator onto Stack
	Operation:	\Downarrow (A); SP \leftarrow (SP-\$01)
	Description:	The contents of the accumulator are pushed onto the stack at the address contained in the stack pointer. The stack pointer is then decremented to point at the next available location in the stack. The contents of the accumulator remain unchanged.
PSHH	Push Index F	Register H onto Stack
	Operation:	\Downarrow (H); SP \leftarrow (SP-\$01)
	Description:	The contents of the 8-bit high order index register H are pushed onto the stack at the address contained in the stack pointer. The stack pointer is then decremented to point at the next available location in the stack. The contents of the H register remain unchanged.

PSHX	Operation:	Register X onto Stack \downarrow (X); SP \leftarrow (SP-\$01) The contents of the 8-bit low order index register X are pushed onto the stack at the address contained in the stack pointer. The stack pointer is then decremented to point at the next available location in the stack. The contents of the X register remain unchanged.
PULA		lator from Stack
	Operation:	$SP \leftarrow (SP+\$01); \ \Uparrow (A)$
	Description:	The stack pointer is incremented to address the last operand on the stack. The accumulator is then loaded with the contents of the address pointed to by SP.
PULH	Pull Index Re	egister H from Stack
	Operation:	$SP \leftarrow (SP+\$01); \uparrow (H)$
	Description:	The stack pointer is incremented to address the last operand on the stack. The 8-bit index register H is then loaded with the contents of the address pointed to by SP.
PULX	Pull Index Re	egister X from Stack
	Operation:	$SP \leftarrow (SP+\$01); \ \widehat{1} (X)$
	Description:	The stack pointer is incremented to address the last operand on the stack. The 8-bit index register X is then loaded with the contents of the address pointed to by SP.
тѕх	Transfer Stac	ck Pointer to Index Register
	Operation:	H:X ← (SP) + \$0001
	Description:	Loads the index register H:X with one plus the contents of the 16-bit stack pointer SP. The contents of the stack pointer remain unchanged. After a TSX instruction, the

index register H:X points to the last value that was stored on the stack.

TXSTransfer Index Register to Stack PointerOperation: $SP \leftarrow (H:X) - \$0001$ Description:Loads the stack pointer SP with the
contents of the index register H:X minus
one.The contents of the index register H:X
remain unchanged.

Software
Techniques Using
the SPThe CPU05 and the CPU08 use the stack for two primary purposes.
First, every time the CPU executes an interrupt service routine, the
register contents are saved on the stack. After the execution of a return
from interrupt (RTI) instruction, the register contents on the stack are
restored to the CPU. Second, every time a jump to subroutine (JSR) or
a branch to subroutine (BSR) occurs, the return address is saved on the
stack. The address is restored to the program counter after a return from
subroutine (RTS) instruction is executed.

The CPU08 with its new stack manipulation instructions allows the user to pass parameters to the subroutine and store local or temporary values within the subroutine. Two major benefits are derived from using the stack for parameters and temporary values:

- A subroutine will allocate RAM storage for its variables and release this memory when the subroutine is finished. Therefore, global variables are not needed for these routines. This saves RAM memory space.
- 2. The allocation of new local variables for each subroutine makes the subroutine recursive and reentrant. This allows the programmer to easily modularize his code.


Let's look at the stacking operation of the CPU05 and the CPU08. The stack is located in RAM. Since stacking occurs from high memory to low memory, the SP usually points to the highest RAM memory address. Both the CPU05 and the CPU08 reset the SP at \$00FF. The CPU08 instruction set allows the programmer to move the stack out of Page 0 memory if needed.

Application Note

When an interrupt occurs, the contents of all the CPU registers are pushed onto the stack, the interrupt vector is fetched, and the program begins execution at the start of the interrupt routine. The stack contents before and after an interrupt are shown in **Figure 2**. For the CPU08 to remain upward compatible with the CPU05, the H index register is not pushed onto the stack.

NOTE: If the H register is used in the interrupt service routine or if indexed addressing modes are used, the H register must be pushed onto the stack.

This is accomplished by using the PSHH instruction. Before returning from the interrupt, the PULH instruction must be used to extract the H index register off the stack.



STACK BEFORE AN INTERRUPT CALL







Figure 2. Stack Before and After an Interrupt Call

Figure 3 illustrates the stack before and after a subroutine is called when the stack pointer is at \$00FF. When a subroutine is called, the 16bit program counter is pushed onto the stack and the execution of code begins at the start of the subroutine. The program counter is split into its 8-bit high and low bytes.



STACK BEFORE A SUBROUTINE CALL




Figure 3. Stack Before and After a Subroutine Call

If the values in the X register and the accumulator are needed within a subroutine, they will need to be saved somehow before the subroutine uses them. If using the CPU05, you would have to allocate global RAM space for saving these CPU registers. Your code would look something like that in Figure 4.

*	Subrou	utine XX for	CPU05 *
START	STX	\$50	;4 store X away to RAM
	STA	\$51	;4 store A away to RAM
	XX XX	XX XX	;actual subroutine code
	LDX	\$50	;3 load X from RAM
	LDA	\$51	;3 load A from RAM

Figure 4. CPU05 Subroutine Code

The CPU05 code will use 14 cycles to store and load registers. Also, two bytes of global RAM space are allocated for this subroutine. If we were to use the CPU08, the code could utilize the stack. Global RAM space and six cycles would be saved. Refer to **Figure 5**.

*	Subroutine X	XX for CPU08	*
START	PSHX PSHH PSHA	;2 pu	ash X onto stack ash H onto stack ash A onto stack
	XX XX XX XX	;actu	al subroutine code
	PULA PULH PULX	;2 pu	all A off of stack all H off of stack all X off of stack

Figure 5. CPU08 Subroutine Code

The stack helps in efficiently utilizing parameters, local variables, and subroutine return values. Parameters are variables that are passed to the subroutine. Local variables are variables that are only used within the scope of the subroutine. A subroutine return value is the output of the subroutine. An example of a subroutine and its variables are given below in equation form:

$$Y = (X)3$$

If we were to write a subroutine that calculates the cube of the value X, X would be the parameter passed to the subroutine. Y would be the subroutine return value, and any variable used to calculate Y would be

a local variable. The stack of these complex subroutines follow the generalized structure shown in **Figure 6**. **Figure 6** shows the stack before the subroutine initialization, before entering the subroutine, and during the subroutine. The actual cube subroutine is written in the following section of code. A diagram of the stack during its execution is given within the code listing.

		STACK	
SP DURING SUBROUTINE	\rightarrow		\$00f7
		LOCAL VARIABLE 1	\$00F8
		LOCAL VARIABLE 2	\$00F9
		ACCUMULATOR A	\$00FA
		INDEX REGISTER X	\$00FB
		PROGRAM COUNTER HIGH	\$00FC
SP DURING SUBROUTINE	\rightarrow	PROGRAM COUNTER LOW	\$00FD
		RETURN VALUE	\$00FE
SP BEFORE SUBROUTINE INITIALIZATION	\rightarrow	PASSED PARAMETER	\$00FF

Figure 6. Stack Structure of a Complex Subroutine

Code ExampleRefer to Appendix D — Using the Stack in a Subroutine to Compute
a Cube for an example of modular subroutine code that efficiently
computes the cube of an 8-bit positive number.

Data Movement

Why Improve the Movement of Data in the CPU05?

The most common CPU function is the transfer of data. Most microcontroller-based systems spend the majority of their time moving data from one location to the other. Many different addressing modes are used to access and transfer bytes of data. If there was a way to decrease the time it takes to transfer data, then the overall performance of the system would be improved.

CPU05 moves data from one location to the next by first loading the accumulator with the byte from the transfer source. Next, CPU05 stores the byte from the accumulator to the transfer's destination. In this manner all data must pass through the accumulator, thus making the accumulator a bottleneck in data movement. The movement of the contents of location \$40 to location \$60 with the CPU05 is illustrated in **Figure 7**.




Figure 7. Accumulator as a Bottleneck

CPU08 provides the new MOV instruction which bypasses the accumulator. Using the MOV instruction, the CPU is instructed to take the contents of the source location and directly place the data in the destination. This is illustrated in **Figure 8**. There are four different addressing modes special to the MOV instruction. Details of this instruction are given below.




Figure 8. No Accumulator Bottleneck

New MOVE Instruction, Detail	The new CPU08 Move instruction is detailed below. Examples for this instruction and all four of its addressing modes are given in Appendix E — New CPU08 MOV Instruction Examples . The examples in Appendix E — New CPU08 MOV Instruction Examples also compare the CPU05 and the CPU08 bus cycles and memory requirements for the algorithm to execute the movement of data.			
	MOV	Move		
		Operation:	(M)destination \leftarrow (M)source	
		Description:	Moves a byte of data from a source address to a destination address. Data is examined as it is moved, and condition codes are set. Source data is not changed. Internal registers (other than CCR) are not affected. There are four addressing modes for the MOV instruction. A discussion of these modes was given in an earlier section.	
Software Techniques	data. The mo	ost obvious ad	It cycle time and code space when moving dvantage of the MOV instruction is when the being initialized along with other RAM program.	
Code Example	A user wants to start his application one of two different ways. The user initializes the application on the MCU based on the logic level of port D bit 4. Once the part is out of reset, it reads port D and moves data from ROM into the RAM configuration registers according to the logic level of bit 4. Refer to Appendix F — CPU05 and CPU08 Data Movement Code for code comparing the CPU05 and the CPU08.			

New Branch Instructions

Description	of the CPUC These instru DBNZX. The instructions instructions instructions instructions	08 to improve I uctions are CB le CBEQ instru and the branc combine the c and the branc	instructions were added to the instruction set ooping and table searching capabilities. EQ, CBEQA, CBEQX, DBNZ, DBNZA, and actions combine the compare (CMP and CPX) h if equal (BEQ) instruction . The DBNZ lecrement (DEC, DECA, and DECX) h if not equal (BNE) instruction. These new time and decrease code space. More detail struction.
Six New Branch Instructions, Detail	Examples for Branch Ins Branch Ins	or these instruction Exan truction Exan truction Exan cycles and me	ctions that affect branching are listed below. ctions are given in Appendix G — New pples . The examples in Appendix G — New pples also compare the CPU05 and the emory requirements for the algorithm to
	CBEQ	Compare an	d Branch if Equal
		Operation:	A) - (M); PC \leftarrow (PC) + \$0003 + Rel if result is \$00
		For IX+ mod	e: (A) - (M);
		$PC \gets (PC) \cdot$	+ \$0002 + Rel, if result is \$00
		Description:	CBEQ compares the operand from memory with the accumulator and causes a branch if the result is zero. This function combines CMP and BEQ for faster table look-up routines.
			The addressing mode CBEQ_IX+ compares the operand addressed by the 16-bit index register H:X to the accumulator and causes a branch if the result is zero. The 16-bit index register is then incremented regardless of whether a branch is taken. CBEQ_IX1+ operates the same way except

an 8-bit offset is added to the effective address of the operand.

CBEQA Compare and Branch if Equal

Operation: (A) - (M); PC \leftarrow (PC) + \$0003 + Rel if result is \$00

- Description: CBEQA compares an immediate operand in memory with the accumulator and causes a branch if the result is zero. This instruction combines CMP and BEQ for faster table look-up routines.
- **CBEQX** Compare and Branch if Equal
 - Operation: (IX) (M); PC \leftarrow (PC) + \$0003 + Rel if result is \$00
 - Description: CBEQX compares an immediate operand in memory with the lower order index register X and causes a branch if the result is zero. This instruction combines CPX and BEQ for faster loop counter control.

DBNZ Decrement and Branch if Not Zero

- Operation: $M \leftarrow (M) \$01;$
- $$\label{eq:PC} \begin{split} \text{PC} \leftarrow (\text{PC}) + \$0003 + \text{Rel, if result} _\$00 \text{ for Direct, IX1,} \\ & \text{and SP1} \end{split}$$
- $PC \leftarrow (PC) + \$0002 + Rel$, if result _ \$00 for IX
- Description: DBNZ subtracts one from the operand M in memory and causes a branch if the result is not zero. This instruction combines DEC and BNE for faster loop counter control.

DBNZA Decrement and Branch if Not Zero

- Operation: $A \leftarrow (A) \$01$; $PC \leftarrow (PC) + \$0002 + Rel$, if result _ \$00
- Description: DBNZA subtracts one from the accumulator and causes a branch if the result is not zero. This instruction combines DECA and BNE for faster loop counter control.

Code Example

DBNZX	Decrement a	nd Branch if Not Zero
	Operation:	$X \leftarrow (X)$ - \$01; PC \leftarrow (PC) + \$0002 + Rel, if result _ \$00
	Description:	DBNZX subtracts one from the lower index register and causes a branch if the result is not zero. This instruction combines DECX and BNE for faster loop counter control.
counting rou	tines. Compa	ructions can cut cycle time in looping or re and branch routines can be used to search or variable locations. Decrement and branch

The following piece of code shows how the compare and branch instruction searches a table for a match. As an example, let's say that you recently read in a table of 80 A/D data bytes. You would like to know if the signal was saturated above the rails of the A/D converter. You would then search the table for the value \$FF. If found, your code would branch out and execute some control algorithm to attenuate the analog signal. Refer to **Appendix H — CPU05 and CPU08 Search Code** for a comparison of CPU05 and CPU08 code.

Mathematical Operations

V Bit, DIV, DAA, and the NSA Instruction	New features and instructions added to the CPU08 have made some mathematical computations easier. The V bit is added to the CCR to support signed arithmetic. CPU08 has the capability of 16-bit division. The DIV instruction will divide a 16-bit dividend by an 8-bit divisor. For binary coded decimal operations, the CPU08 has a decimal adjust accumulator, DAA, instruction and a nibble swap accumulator, NSA, instruction.
Signed Math and	The V bit in the CCR adds greater programming flexibility to the user.
Signed Branches	The addition of two's complement comparisons can aid in the branching

routines can be used for keeping count in loops.

operations of high level languages such as C. Also, the representation of signed numbers and their operations can easily be computed. This can be especially helpful with digital signal processing algorithms and the proper storage of signed analog to digital readings.

Four New Signed Branch Instructions, Detail

All the new CPU08 instructions that affect signed branching are listed below. Examples for these instructions are given in Appendix I — New CPU08 Signed Branch Instruction Examples.

BGE	Branch if Gre	eater Than or Equal (signed operands)
	Operation:	$PC \leftarrow (PC) + $0002 + Rel, if (N \oplus V)=0,$ i.e., if (A) _ (M), ("signed" numbers)
	Description:	If the BGE instruction is executed immediately after execution of any of the compare or subtract instructions, the branch will occur if and only if the two's complement number represented by the appropriate internal register (A, X, or H:X) was greater than or equal to the two's complement number represented by M.
BGT	Branch if Gre	eater Than (signed operands)
	Operation:	$\label{eq:PC} \begin{array}{l} \leftarrow (PC) + \$0002 + Rel, \mbox{ if } Z+(N \oplus V)=0, \\ \mbox{i.e., if (A) > (M), ("signed" numbers)} \end{array}$
	Description:	If the BGT instruction is executed immediately after execution of any of the compare or subtract instructions, the branch will occur if and only if the two's complement number represented by the appropriate internal register (A, X, or H:X) was greater than the two's complement number represented by M.

BLE	Branch if Les	s Than or Equal (signed operands)
	Operation:	$PC \leftarrow (PC) + 0002 + Rel, \text{ if } Z+(N \oplus V)=1$ i.e., if (A) _ (M), ("signed" numbers)
	Description:	If the BLE instruction is executed immediately after execution of any of the compare or subtract instructions, the branch will occur if and only if the two's complement number represented by the appropriate internal register (A, X, or H:X) was less than or equal to the two's complement number represented by M.
BLT	Branch if Les	s Than (signed operands)
	Operation:	$PC \leftarrow (PC) + $0002 + Rel, if (N \oplus V)=1$ i.e., if (A) < (M), ("signed" numbers)
	Description:	If the BLT instruction is executed immediately after execution of any of the compare or subtract instructions, the branch will occur if and only if the two's complement number represented by the appropriate internal register (A, X, or H:X) was less than the two's complement number represented by M.

New DIV Instruction

The Divide instruction on the CPU08 does not require the lengthy code needed to divide numbers on the CPU05. A description of the Divide instruction is given below. Appendix J — Five Miscellaneous CPU08 Instructions Including BCD, Divide, and CCR Operations shows a short example of using the new Divide instruction. Appendix K — CPU08 Averaging Code illustrates an averaging routine implementing the Divide instruction.

DIV	Divide	
	Operation:	(H:A) / X \rightarrow A; Remainder \rightarrow H
	Description:	Divides a 16-bit unsigned dividend
		contained in the concatenated registers H
		and A by an 8-bit divisor contained in index

register X. The quotient is placed in the accumulator A, and the remainder is placed in the high order index register H. The divisor is left unchanged.

New DAA and the NSA instruction

The decimal adjust accumulator, DAA, and the nibble swap accumulator, NSA, are new instructions to help with binary coded decimal (BCD) operations. The DAA instruction allows the user to adjust the accumulator so that the number represents a BCD number. Swapping nibbles is needed for packing BCD numbers into memory. One use of BCD is data instrumentation. It is easier to store and manipulate these numbers in BCD rather than convert or decode numbers from hexadecimal. Packing is used to store decimal numbers into memory. Instead of one byte storing one decimal, the NSA instruction easily swaps nibbles in the accumulator so that two decimal numbers can be stored in one byte. Appendix J — Five Miscellaneous CPU08 Instructions Including BCD, Divide, and CCR Operations gives examples using the DAA instruction and the NSA instruction. Refer to Appendix L — CPU08 BCD Example Code for an example of BCD code.

DAA Decimal Adjust Accumulator

Operation: (A)10

Description: Adjusts the contents of the accumulator and the state of the CCR carry bit after binary coded decimal operations so that there is a correct BCD sum and an accurate carry indication. The state of the CCR half carry bit affects operation.

NSA Nibble Swap Accumulator
 Operation: A ← (A[3:0]:A[7:4])
 Description: Swaps upper and lower nibbles (4 bits) of the accumulator. This is used for more efficient storage and use of binary coded

operands.

New TAP and TPAThe transfer accumulator to the condition code register, TAP, and the
transfer condition code register to accumulator, TPA, are new
instructions to modify or manipulate the condition code register, CCR.
These instructions are detailed below. Code examples can be found in
Appendix J — Five Miscellaneous CPU08 Instructions Including
BCD, Divide, and CCR Operations.

TAP	Transfer Acc	umulator to Condition Code Register
	Operation:	$CCR \leftarrow (A)$
	Description:	Transfers the contents of the Accumulator to the Condition Code Register.
TPA	Transfer Con	dition Code Register to Accumulator
TPA	Transfer Con Operation:	0

Instruction Cycle Improvements

The CPU08 instruction set not only has new instructions but many of the old instructions are faster. The CPU08 gathers data in a pipeline fashion. Instead of waiting for the instruction to be finished to gather the next opcode or operand, the CPU will fetch the next address byte during the execution of the current instruction. This pipelining overlaps execution of most instructions and thus increases the performance of the CPU08. A list of instructions that were improved is given in Table 3. Please refer to the CPU08 opcode map for further details.

Opcode Mnemonic	Address Mode	HC05 Cycles	HC08 Cycles
ADC	IX	3	2
ADC	IX1	4	3
ADC	IX2	5	4
ADD	IX	3	2
ADD	IX1	4	3
ADD	IX2	5	4
AND	IX	3	2
AND	IX1	4	3
AND	IX2	5	4
ASR	DIR	5	4
ASR	IX	5	3
ASR	IX1	6	4
ASRA	INH	3	1
ASRX	INH	3	1
BCLR0	DIR	5	4
BCLR1	DIR	5	4
BCLR2	DIR	5	4
BCLR3	DIR	5	4
BCLR4	DIR	5	4
BCLR5	DIR	5	4
BCLR6	DIR	5	4
BCLR7	DIR	5	4
BIT	IX	3	2
BIT	IX1	4	3
BIT	IX2	5	4
BSET0	DIR	5	4
BSET1	DIR	5	4
BSET2	DIR	5	4
BSET3	DIR	5	4
BSET4	DIR	5	4
BSET5	DIR	5	4
BSET6	DIR	5	4
BSET7	DIR	5	4
BSR	REL	6	4
CLC	INH	2	1

Table 3. Instruction List (Sheet 1 of 4)

Opcode Mnemonic	Address Mode	HC05 Cycles	HC08 Cycles
CLR	DIR	5	3
CLR	IX	5	2
CLR	IX1	6	3
CLRA	INH	3	1
CLRX	INH	3	1
CMP	IX	3	2
CMP	IX1	4	3
CMP	IX2	5	4
СОМ	DIR	5	4
СОМ	IX	5	3
СОМ	IX1	6	4
COMA	INH	3	1
COMX	INH	3	1
CPX	IX	3	2
СРХ	IX1	4	3
СРХ	IX2	5	4
DEC	DIR	5	4
DEC	IX	5	3
DEC	IX1	6	4
DECA	INH	3	1
DECX	INH	3	1
EOR	IX	3	2
EOR	IX1	4	3
EOR	IX2	5	4
INC	DIR	5	4
INC	IX	5	3
INC	IX1	6	4
INCA	INH	3	1
INCX	INH	3	1
JSR	DIR	5	4
JSR	EXT	6	5
JSR	IX	5	4
JSR	IX1	6	5
JSR	IX2	7	6
LDA	IX	3	2

Table 3. Instruction List (Sheet 2 of 4)

Opcode Mnemonic	Address Mode	HC05 Cycles	HC08 Cycles
LDA	IX1	4	3
LDA	IX2	5	4
LDX	IX	3	2
LDX	IX1	4	3
LDX	IX2	5	4
LSL	DIR	5	4
LSL	IX	5	3
LSL	IX1	6	4
LSLA	INH	3	1
LSLX	INH	3	1
LSR	DIR	5	4
LSR	IX	5	3
LSR	IX1	6	4
LSRA	INH	3	1
LSRX	INH	3	1
MUL	INH	11	5
NEG	DIR	5	4
NEG	IX	5	3
NEG	IX1	6	4
NEGA	INH	3	1
NEGX	INH	3	1
NOP	INH	2	1
ORA	IX	3	2
ORA	IX1	4	3
ORA	IX2	5	4
ROL	DIR	5	4
ROL	IX	5	3
ROL	IX1	6	4
ROLA	INH	3	1
ROLX	INH	3	1
ROR	DIR	5	4
ROR	IX	5	3
ROR	IX1	6	4
RORA	INH	3	1
RORX	INH	3	1

Table 3. Instruction List (Sheet 3 of 4)

Opcode Mnemonic	Address Mode	HC05 Cycles	HC08 Cycles
RSP	INH	2	1
RTI	INH	9	7
RTS	INH	6	4
SBC	IX	3	2
SBC	IX1	4	3
SBC	IX2	5	4
SEC	INH	2	1
STA	DIR	4	3
STA	EXT	5	4
STA	IX	4	2
STA	IX1	5	3
STA	IX2	6	4
STOP	INH	2	1
STX	DIR	4	3
STX	EXT	5	4
STX	IX	4	2
STX	IX1	5	3
STX	IX2	6	4
SUB	IX	3	2
SUB	IX1	4	3
SUB	IX2	5	4
SWI	INH	10	9
TAX	INH	2	1
TST	DIR	4	3
TST	IX	4	2
TST	IX1	5	3
TSTA	INH	3	1
TSTX	INH	3	1
TXA	INH	2	1
WAIT	INH	2	1

Table 3. Instruction List (Sheet 4 of 4)

Conclusion

This application note has covered the differences between the HC05 and the HC08 CPU architecture. Please refer to the *M68HC05 Applications Guide* for further study of the CPU05. The *CPU08 Reference Manual* is a valuable resource for studying the CPU08 in more detail.

Please consult your local Motorola sales office or your authorized Motorola distributor for applications support, literature, and specific part information.

The MCU BBS is also available with free software for use with HC05 and HC08 MCUs. The BBS number is (512) 891-3733. The code examples used in this application note can be found on the BBS. The file name is HC08OPT.ARC.

Appendix A — New CPU08 Indexing Instruction Examples

```
File : INDEX.ASM
       Description :
              Shows examples for new CPU08 indexing
               instructions - AIX, CLRH, CPHX, LDHX, STHX
              Not all addressing modes are shown.
       Note : Please consult the CPU08 Reference Manual
               for further details on these instructions
              Code is written for educational
               purposes only
  ORG
              $200
* * * * *
       AIX - add immediate to index register
START
       LDHX
              #$1010
                             ; H:X ← $1010
                             ; H:X = $1010 + (-$10)
       AIX
              #-10
                                 = $1000
                             ;
* * * * *
       CLRH - clear index high
              #$1290
                             ; H:X ← $1290
       LDHX
                             ; H:X ← $0090
       CLRH
* * * * *
       CPHX - compare 16-bit index register
       LDHX
              #$1290
                             ; H:X ← $1290
                             ; CCR = %0110,1000
                             ; CCR before CPHX, Z=0
       CPHX
              #$1290
                             ; H:X ← $1290
                             ; CCR = %0110,1010
                             ; CCR after CPHX, Z=1
* * * * *
       LDHX - load 16-bit index register
       LDHX
              #$1290
                             ; H:X ← $1290
* * * * *
       STHX - store 16-bit index register
              #$1290
                             ; H:X ← $1290
       LDHX
       STHX
              $50
                             ; ($50) ← (H:X)
                             ; ($50) ← $12
                             ; ($51) ← $90
DONE
       NOP
       BRA
              DONE
* * * * *
       Initialize the reset vector
       ORG
              $FFFE
       DW
              START
```

Appendix B — CPU05 and CPU08 512-Byte Table Indexing Code

* * * * * * * *	* * * * * * * *	* * * * * * * * * * * * * * * * *	* * * *	* * * * * * * * * * * * * * * * * * * *
*				
*	File :	INDEXX.ASM		
*	Descript	tion :		
*		The following c	ode	illustrates the
*		different inst	ruct	ions used to address
*		a 512 byte tab	le i	in memory. HC05 and HC08
*		code is compar	ed.	
*	Notes:	Comments to the	rig	ght of some instructions
*		give numbers.		
*		CPU05 - 1st # i	s CI	2005 cycle count
*		2nd # i	s ir	nstruction byte count
*		CPU08 - 1st # i	s CI	2008 cycle count
*				struction byte count
*				CPU08 Reference Manual
*		for further de	tail	ls on these instructions
*		Code is written	foi	c educational
*		purposes only		
*				
* *		predefined in R		ample, the table address
TBL_ST0 TBL_ST1		\$400 TBL_ST0+256T		start of table, section (start of table, section 2
	ORG	\$50	;	start of RAM variables
TBL A	RMB	2	;	
100_11		-	;	accessed by the code
	ORG	\$200		

*	***************************************			
*	ACUS CODE			
*	CPU05 has to address the table in a section-like fashion. Section 0 is between \$400 and \$4FF. Section 1 is between \$500 and \$5FF. The 16-bit address is stored in RAM location TBL_A. This is the offset to the table starting at \$400, TBL_STO. Example: Address is \$520 = \$400 + \$120 TBL_A = \$01			
*				
*				
*				
*				
*				
*				
*		TBL_A+1 = \$20		
START	LDX	TBL_A+1	;3,2 X ← (TBL_A+1)	
	LDA	TBL_A	;3,2 A \leftarrow (TBL_A)	
	BEQ	TBL0	;3,2 branch to section 0 if	
	LDA	TBL_ST1,X	;5,3 A \leftarrow (X+TBL_ST1)	
	BRA	NEXT	;3,2 branch when done to	
			; the CPU08 example	
TBL0	LDA	TBL_ST0,X	;5,3 A \leftarrow (X+TBL_ST0)	
*	Total #	CPU05 cycles		
*	Total #	bytes	= 11 (max)	
* * * * * * * * *	HC08 co		*** * indexed addressing so the	
*	table		ded from TBL_A in RAM. No	
NEXT	LDHX	TBL A	;4,2 H:X \leftarrow (TBL A)	
	LDA	_ TBL_ST0,X		
*	Total #	CPU08 cycles	= 8	
*	Total #	bytes	= 5	
* * * * * * *	* * * * * * * * *	* * * * * * * * * * * * * * * *	* * *	
DONE	NOP			
	BRA	DONE		
		nitialize the reset vector		
* * * * *			ector	
* * * * *	Initial ORG DW	ize the reset v \$FFFE START	ector	

Appendix C — New CPU08 Stack Pointer Instructions

```
File : SP.ASM
       Description :
               Shows examples for new CPU08 stack pointer
                instructions - AIS, PSHA, PSHH, PSHX
                              PULA, PULH, PULX, TSX, TXS
              Not all addressing modes are shown.
*
       Note :
              Please consult the CPU08 Reference Manual
                for further details on these instructions
*
               Code is written for educational
               purposes only
ORG
              $200
* * * * *
       AIS - add immediate to stack pointer
       SP is predefined at $0FE0
START
              #$1F
                              ; SP ← $0FE0 + $1F
       AIS
                              ; SP = \$0FFF
****
       PSHA - push accumulator onto stack
       SP is predefined at $0FFF
*
       A = $80
       PSHA
                              ; ($0FFF) ← $80
                              ; SP ← SP-$01
                              ; SP = $0FFE
* * * * *
       PSHH - push index register H onto stack
       SP is predefined at $0FFE
       H:X = $2050
       PSHH
                              ; ($0FFE) ← $20
                              ; SP ← SP-$01
                              ; SP = $0FFD
* * * * *
       PSHX - push index register X onto stack
       SP is predefined at $0FFD
       H:X = $2050
       PSHX
                              ; (\$0FFD) = \$50
                              ; SP \leftarrow SP-$01
                              ; SP = \$0FFC
```

```
* * * * *
        PULX - pull index register X from stack
*
        SP is predefined at $0FFC
*
        $0FFD = $50
*
        H:X = $0000
        PULX
                                   ; SP ← SP+$01
                                   ; SP = \$0FFD
                                   ; X \leftarrow (\$0FFD)
                                   ; H:X = $0050
* * * * *
        PULH - pull index register H from stack
*
        SP is predefined at $0FFD
*
        $0FFE = $20
*
        H:X = $0050
        PULH
                                   ; SP ← SP+$01
                                   ; SP = $0FFE
                                   ; H \leftarrow (\$0FFE)
                                   ; H:X = $2050
* * * * *
        PULA - pull accumulator from stack
*
        SP is predefined at $0FFE
*
        $0FFF = $80
*
        A = $00
        PULA
                                   ; SP ← SP+$01
                                   ; SP = \$0FFF
                                   ; A \leftarrow ($0FFF)
                                   ; A = $80
* * * * *
        TSX - transfer stack pointer to index register
*
        SP is predefined at $0FF5
*
        H:X = $1290
        TSX
                                   ; H:X ← SP+$01
                                   ; H:X = $0FF6
* * * * *
        TXS - transfer index register to stack pointer
*
        SP is predefined at $0FF5
*
        H:X = $1290
        TXS
                                   ; SP ← H:X-$01
                                   ; SP = $128F
DONE
        NOP
        BRA
                 DONE
* * * * *
        Initialize the reset vector
        ORG
                 $FFFE
        DW
                 START
```

Appendix D — Using the Stack in a Subroutine to Compute a Cube

* * * * * * *	******	* * * * * * * * * * * * * * * * * * * *				
*						
*	File :	CUBE.ASM				
*	Descrip	cription :				
*		This program takes an 8-bit positive				
*		number, X_IN, and cubes it. The answer,				
*		Y_IN, is in a 24-bit format.				
*		This program also illustrates the				
*		value of using the stack for complex				
*		subroutines that use parameter passing,				
*		local variables, and return values.				
*	Stack D	escription:				
*		Given below is a diagram of the stack				
*		during the subroutine				
*		The numbers on the right specify the				
*		number of bytes above the stack pointer				
*						
*						
*		$SP \rightarrow ??$				
*		VAR1 1				
*		VAR2 2				
*		A 3				
*		н 4				
*		x 5				
*		PC HIGH 6				
*		PC_LOW 7				
*		Y HIGH 8				
*		Y MED 9				
*		Y_LOW 10				
*		X_IN 11				
*		··_···				
*						
*	Note :	Please consult the CPU08 Reference Manual				
*	1.000	for further HC08 instruction details				
*		Code is written for educational				
*		purposes only				
*						
* * * * * * *	******	*****				
	ORG	\$80				
X_IN	RMB	1 ;8-bit number to be cubed				
	ORG	\$200				

* * * * *	Load up stack before entering the subroutine					
*			t number to be cubed, X_IN			
*			ade available to the stack			
*	for the 24 bit output of the routine					
*		3 pushes are made to illustrate this point				
	5 pusiies	are made to II	iustrate this point			
~~~~~~						
START		X_IN	$; A \leftarrow (X_{IN})$			
	PSHA		;push parameter X_IN onto stack			
	CLRA		;zero must be pushed on stack			
			; allocation for return answer			
	PSHA		;push Y_Low byte onto stack			
	PSHA		;push Y_Med byte onto stack			
	PSHA		;push Y_High byte onto stack			
* * * * *	Jump to	the cube subrou	tine			
	-	CUBE	;jump sub to CUBE, Y = X_IN^3			
			· J ·····			
* * * * *	When subr	routine is over.	reset stack pointer to original			
*			wers off the stack when needed.			
	rocación	· I uII chie und	weib off the Stack when needed.			
	AIS	#\$04	;SP ← (SP) + \$04			
	ALD	πφ01				
	BRA	DONE	;branch to the end of this			
	DRA .	DONE				
			;example			
* * * * *	CUDE cub	moutine				
* * * * *	CUBE sub		2			
* * * * * * * * * *		proutine IN, find Y = X^	3			
* * * * *	Given X_	IN, find $Y = X^{*}$				
* * * * *	Given X_ Save X,H	IN, find $Y = X^{*}$ , and A on stack	k			
* * * * * * *	Given X_ Save X,H Decremen	IN, find $Y = X^{*}$	k ytes			
* * * * *	Given X_ Save X,H	IN, find $Y = X^{*}$ , and A on stack	k ytes ;push X onto stack			
* * * * * * *	Given X_ Save X,H Decremen	IN, find $Y = X^{*}$ , and A on stack	k ytes ;push X onto stack ;push H onto stack			
* * * * * * *	Given X_ Save X,H Decremen PSHX	IN, find $Y = X^{*}$ , and A on stack	k ytes ;push X onto stack			
* * * * * * *	Given X_ Save X,H Decremen PSHX PSHH PSHA	IN, find $Y = X^{*}$ , and A on stack	k ytes ;push X onto stack ;push H onto stack			
* * * * * * *	Given X_ Save X,H Decremen PSHX PSHH PSHA	IN, find Y = X [*] , and A on stac t stack for 2 b	k ytes ;push X onto stack ;push H onto stack ;push A onto stack			
* * * * * * *	Given X_ Save X,H Decremen PSHX PSHH PSHA AIS	IN, find Y = X [*] , and A on stac t stack for 2 b	k ytes ;push X onto stack ;push H onto stack ;push A onto stack			
***** * CUBE	Given X_ Save X,H Decremen PSHX PSHH PSHA AIS Run the r	IN, find Y = X [^] I, and A on stac It stack for 2 b #-2	k ytes ;push X onto stack ;push H onto stack ;push A onto stack ;decrement stack for local var			
***** * CUBE	Given X_ Save X,H Decremen PSHX PSHH PSHA AIS Run the S Square X	IN, find Y = X [^] , and A on stac t stack for 2 b #-2 math routine _IN, answer is 1	k ytes ;push X onto stack ;push H onto stack ;push A onto stack ;decrement stack for local var X:A			
***** * CUBE	Given X_ Save X,H Decremen PSHX PSHH PSHA AIS Run the T Square X LDA	IN, find Y = X [^] , and A on stack t stack for 2 b #-2 math routine _IN, answer is 1 11T,SP	k ytes ;push X onto stack ;push H onto stack ;push A onto stack ;decrement stack for local var X:A ;A = X_IN			
***** * CUBE	Given X_ Save X,H Decremen PSHX PSHH PSHA AIS Run the Square X LDA LDX	IN, find Y = X [^] , and A on stac t stack for 2 b #-2 math routine _IN, answer is 1	k ytes ;push X onto stack ;push H onto stack ;push A onto stack ;decrement stack for local var X:A ;A = X_IN ;X = X_IN			
***** * CUBE	Given X_ Save X,H Decremen PSHX PSHH PSHA AIS Run the T Square X LDA	IN, find Y = X [^] , and A on stack t stack for 2 b #-2 math routine _IN, answer is 1 11T,SP	k ytes ;push X onto stack ;push H onto stack ;push A onto stack ;decrement stack for local var X:A ;A = X_IN			
***** * CUBE	Given X_ Save X,H Decremen PSHX PSHH PSHA AIS Run the T Square X LDA LDX MUL	IN, find Y = X ⁴ , and A on stack t stack for 2 by #-2 math routine _IN, answer is 1 11T,SP 11T,SP	k ytes ;push X onto stack ;push H onto stack ;decrement stack for local var X:A ;A = X_IN ;X = X_IN ;X:A = (X)*(A)			
***** * CUBE *	Given X_ Save X,H Decremen PSHX PSHH PSHA AIS Run the r Square X LDA LDX MUL Store aw	<pre>IN, find Y = X^ A, and A on stach t stack for 2 by #-2 math routine LIN, answer is A 11T,SP 11T,SP ray the high byt</pre>	<pre>k ytes ;push X onto stack ;push H onto stack ;push A onto stack ;decrement stack for local var X:A ;A = X_IN ;X = X_IN ;X:A = (X)*(A) e answer, X, to var1</pre>			
***** * CUBE *	Given X_ Save X,H Decremen PSHX PSHH PSHA AIS Run the r Square X LDA LDX MUL Store aw	IN, find Y = X ⁴ , and A on stack t stack for 2 by #-2 math routine _IN, answer is 1 11T,SP 11T,SP	k ytes ;push X onto stack ;push H onto stack ;decrement stack for local var X:A ;A = X_IN ;X = X_IN ;X:A = (X)*(A)			
***** * CUBE *	Given X_ Save X,H Decremen PSHX PSHH PSHA AIS Run the T Square X LDA LDX MUL Store aw STX	<pre>IN, find Y = X^ , and A on stac. t stack for 2 b; #-2 math routine _IN, answer is : 11T,SP 11T,SP ray the high byt 1,SP</pre>	k ytes ;push X onto stack ;push H onto stack ;push A onto stack ;decrement stack for local var X:A ;A = X_IN ;X = X_IN ;X:A = (X)*(A) e answer, X, to varl ;store high answ to varl			
***** * CUBE * *	Given X_ Save X,H Decremen PSHX PSHH PSHA AIS Run the T Square X LDA LDX MUL Store aw STX Multiply	IN, find Y = X ⁴ , and A on stack t stack for 2 by #-2 math routine IN, answer is 1 11T,SP 11T,SP ray the high byt 1,SP 16 bit result 1	<pre>k ytes ;push X onto stack ;push H onto stack ;push A onto stack ;decrement stack for local var X:A ;A = X_IN ;X = X_IN ;X:A = (X)*(A) e answer, X, to varl ;store high answ to varl by X_IN</pre>			
***** * CUBE * *	Given X_ Save X,H Decremen PSHX PSHH PSHA AIS Run the T Square X LDA LDX MUL Store aw STX Multiply Multiply	<pre>IN, find Y = X^ i, and A on stach t stack for 2 by #-2 math routine [_IN, answer is 1 IT,SP 11T,SP ray the high byt 1,SP fild bit result 1 T</pre>	<pre>k ytes ;push X onto stack ;push H onto stack ;push A onto stack ;decrement stack for local var X:A ;A = X_IN ;X = X_IN ;X:A = (X)*(A) e answer, X, to var1 ;store high answ to var1 by X_IN te of 16-bit square</pre>			
***** * CUBE * *	Given X_ Save X,H Decremen PSHX PSHH PSHA AIS Run the T Square X LDA LDX MUL Store aw STX Multiply Multiply	IN, find Y = X ⁴ , and A on stack t stack for 2 by #-2 math routine IN, answer is 1 11T,SP 11T,SP ray the high byt 1,SP 16 bit result 1	<pre>k ytes ;push X onto stack ;push H onto stack ;push A onto stack ;decrement stack for local var X:A ;A = X_IN ;X = X_IN ;X:A = (X)*(A) e answer, X, to varl ;store high answ to varl by X_IN</pre>			

#### Application Note Appendix D — Using the Stack in a Subroutine to Compute a Cube

*	Store a	away low byte of	16-bit result
*	to Y_		
*	Store ł		bit result to var2
	STA	10T,SP	;store low answ to Y_LOW
	STX	2,SP	;store high answ to var2
*	Maa ] + daa ]	le bish hets of 1	( bit we sult be V TN
~			6-bit result by X_IN
	LDA	11T,SP	$A \leftarrow X_{IN}$
	LDX MUL	1,SP	;load X with varl ;X:A = X IN * varl
	моц		/X·A - X_IN " Vali
*	Store h	nigh byte of answ	ver to Y HIGH
	STX	8T,SP	;store high byte to Y_HIGH
*	ADD vai	2 to the low byt	e answer to get Y_MED
*	If the	re is a carry, ad	ld one bit to Y_HIGH
	ADD	2T,SP	A = var2 + A
	BCS	CS	;branch if C bit set in CCR
	BRA	FIN	;C bit is 0, branch to FIN
CS	INC	8T,SP	;add 1 to Y_HIGH
ETN	STA	9T,SP	STARA A TO Y MED
FIN	SIA	91,SP	;store A to Y_MED
*	Save X	H, and A on stac	2k
*	Increme	ent stack for 2 k	bytes
*	Restore	X,H, and A	
*	Return	from the subrout	line
	AIS	#\$02	
	PULA		
	PULH		
	PULX		
	RTS		
DONE	NOP		
	BRA	DONE	
* * * * *	Tni+ial	lize the reset ve	ator
	ORG	\$FFFE	CLUI
	DW	START	

Appendix E — New CPU08 MOV Instruction Examples

*	File : MOVE.ASM		
*			
*	Description : Shows examples for the MOV instruction		
*	All four addressing modes are illustrated	J	
*	_	ı	
^ *	05 and 08 code is compared		
	Notes: Comments to the right of some instruction	าร	
*	give numbers.		
*	CPU05 - 1st # is CPU05 cycle count		
*	2nd # is instruction byte count		
*	CPU08 - 1st # is CPU08 cycle count		
*	2nd # is instruction byte count		
*	Please consult the CPU08 Reference Manua		
*	for further details on these instruction	າຄ	
*	Code is written for educational		
*	purposes only		
*			
	ORG \$200		
* * * * *	Move Immediate value to Direct memory location		
*	нс05 *		
START	LDA $\#$ \$30 ;2,2 A $\leftarrow$ \$30		
	STA $\$80$ ;4,2 ( $\$80$ ) $\leftarrow$ (A)		
*	HC08 *		
	MOV $\#$ \$30,\$80 ;4,3 (\$80) $\leftarrow$ \$30		
*	Total CPU05 cycles, bytes = 6,4		
*	Total CPU08 cycles, bytes = 4,3		
* * * * *	Move Direct mem value to Direct mem location		
*	HC05 *		
	LDA $\$80$ ;3,2 A $\leftarrow$ ( $\$80$ )		
	STA $\$90$ ;4,2 ( $\$90$ ) $\leftarrow$ (A)		
*	HC08 *		
	MOV $$80,$90$ ;5,3 (\$90) $\leftarrow$ (\$80)		
*	MOV $\$80,\$90$ ;5,3 ( $\$90$ ) $\leftarrow$ ( $\$80$ ) Total CPU05 cycles, bytes = 7,4		

* * * * *	Move contents of Indexed Xreg	to Direct mem location, post inc
*	HC05 LDX #\$80 LDA ,X STA \$90 INCX	* ; $X \leftarrow \$80$ ;3,1 $A \leftarrow (X)$ ;4,2 ( $\$90$ ) $\leftarrow$ (A) ;3,1 $X \leftarrow X + 1$
*	HC08 LDX #\$80 MOV X+,\$90	* ; $X \leftarrow \$80$ ;4,2 ( $\$90$ ) $\leftarrow$ (X) ; $X \leftarrow X + 1$
*	Total CPU05 cycles, byte	
*	Total CPU08 cycles, byte	es = 4,2
* * * * *	Move Direct mem contents Xreg	s to Indexed location, post inc
*	НС05	*
	LDX #\$90	; X ← \$90
	LDA \$80	;3,2 A ← (\$80)
	STA ,X	$;4,1$ (X) $\leftarrow$ (A)
	INCX	$;3,1 X \leftarrow X + 1$
*	HC08	*
	LDX #\$90	; X ← \$90
	MOV \$80,X+	;4,2 (X) ← (\$80)
		; $X \leftarrow X + 1$
*	Total CPU05 cycles, byte	r = 10.4
*	Total CPU08 cycles, byte	
* * * * *	Initialize the reset vec ORG \$FFFE DW START	etor

Appendix F — CPU05 and CPU08 Data Movement Code

```
******
       File : MOVEX.ASM
*
       Description :
              A user wants to start an application one of
               two different ways. The user sets the
               application on the MCU by the logic level
               of Port D, bit 3. Once out of reset, the
               MCU reads Port D and moves data from ROM
               into the RAM configuration registers
*
*
               according to the logic level of bit 3.
       Notes: Comments to the right of some instructions
               give numbers.
              CPU05 - 1st # is CPU05 cycle count
                      2nd # is instruction byte count
*
              CPU08 - 1st # is CPU08 cycle count
*
                      2nd # is instruction byte count
              Please consult the CPU08 Reference Manual
               for further details on these instructions
              Code is written for educational
               purposes only
*
*
       For the purpose of this example, we will be using
*
       random ctrl registers for the code. They are listed
       below in an equate table
                           ; start of table
TBL
       EOU
              $1000
PORTD
       EQU
              $03
                            ; port D data register
PORTADR EQU
              $04
                           ; port A data direction register
PORTBDR EQU
              $05
                           ; port B data direction register
SPICTRL EQU
            $0A
                           ; SPI control register
SCICTRL EOU
            $0E
                            ; SCI control register
            $12
TIMCTRL EQU
                            ; Timer control register
       ORG
              $200
* * * * *
       If bit 3 = 0 when read, then the table
*
       starts at $1000
*
       If bit 3 = 1 when read, then the table
       starts at $1008
HC05 code
START05 LDA
             PORTD
                            ;3,2 A \leftarrow (PORTD)
       AND
              #$08
                            ;2,2 clear A except bit 3
                             ; A = 0 \text{ or } 8
                             ;2,1 X \leftarrow (A)
       TAX
                                set the offset of X
                             ;
```

LDA TBL,X ;5,3 A  $\leftarrow$  (TBL+X) STA PORTADR ;4,2 (PORTADR)  $\leftarrow$  (A)  $;3,1 X \leftarrow X + 1$ INCX ;5,3 A  $\leftarrow$  (TBL+X) LDA TBL,X ;4,2 (PORTBDR)  $\leftarrow$  (A) STA PORTBDR INCX  $;3,1 X \leftarrow X + 1$ LDA TBL,X ;5,3 A  $\leftarrow$  (TBL+X) STA SPICTRL ;4,2 (SPICTRL)  $\leftarrow$  (A) ;3,1  $X \leftarrow X + 1$ INCX TBL,X LDA ;5,3 A  $\leftarrow$  (TBL+X) ;4,2 (SCICTRL)  $\leftarrow$  (A) STA SCICTRL  $;3,1 X \leftarrow X + 1$ INCX ;5,3 A  $\leftarrow$  (TBL+X) LDA TBL,X STA TIMCTRL ;4,2 (TIMCTRL)  $\leftarrow$  (A) * Total # CPU05 cycles = 64 Total # bytes = 34 ****** ***** * * HC08 code START08 LDHX #TBL ;3,3 H:X  $\leftarrow$  TBL LDA PORTD ;3,2 A  $\leftarrow$  (PORTD) AND ;2,2 clear A except bit 3 #\$08 ; A = 0 or 8TAX  $;1,1 X \leftarrow (A)$ set the offset of X ; MOV X+,PORTADR ;4,2 (PORTADR)  $\leftarrow$  (H:X) ;  $X \leftarrow X + 1$ MOV X+,PORTBDR ;4,2 (PORTBDR)  $\leftarrow$  (H:X) ;  $X \leftarrow X + 1$ MOV X+,SPICTRL ;4,2 (SPICTRL)  $\leftarrow$  (H:X) ;  $X \leftarrow X + 1$ MOV X+,SCICTRL ;4,2 (SCICTRL)  $\leftarrow$  (H:X) ;  $X \leftarrow X + 1$ X+,TIMCTRL MOV ;4,2 (TIMCTRL)  $\leftarrow$  (H:X) ;  $X \leftarrow X + 1$ = 29 * Total # CPU08 cycles Total # bytes = 18 ****** ***** DONE NOP BRA DONE * * * * * Initialize the reset vector ORG \$FFFE DW START05

# **Application Note**

# Appendix G — New Branch Instruction Examples

*	File :	BRANCH.ASM			
*	Descrip	tion :			
*		Shows examples	for n	ew C	CPU08 branch
*		instructions -	CBEQ,	CBE	EQA, CBEQX
*			DBNZ,	DBN	IZA, DBNZX
*	Notes:		e righ	t of	some instructions
*		give numbers.			
*		CPU05 - 1st # :			
*					tion byte count
*		CPU08 - 1st # :			-
*					tion byte count
*					Reference Manual
*				-	these instructions
^ *		Code is written	n Ior	eauc	ational
*		purposes only			
*****	* * * * * * * * * *	* * * * * * * * * * * * * * * *	* * * * * *	* * * *	* * * * * * * * * * * * * * * * * * * *
	ORG	\$200			
* * * * *	CBEQ -	compare and bra	nch if	equ	al, direct
* * * * * *		compare and bran edefined at \$40	nch if	equ	al, direct
*	A is pr				
	A is pr Memory	edefined at \$40 location \$80 co			
*	A is pro Memory HC05 cod	edefined at \$40 location \$80 con de	ntains *	\$40	)
*	A is pro Memory HC05 coo CMP	edefined at \$40 location \$80 con de \$80	ntains * ;3,2	\$40 (A)	) - (\$80)
*	A is pro Memory HC05 cod	edefined at \$40 location \$80 con de	ntains * ;3,2 ;3,2	\$40 (A) if	) - (\$80) (A) = (\$80) then
*	A is pro Memory HC05 coo CMP	edefined at \$40 location \$80 con de \$80	ntains * ;3,2 ;3,2 ;	\$40 (A) if bra	) - (\$80)
* * LPA	A is pro Memory HC05 coo CMP BEQ BRA	edefined at \$40 location \$80 con de \$80 LP1 LPA	ntains * ;3,2 ;3,2 ; ;	\$40 (A) if bra	) (A) = (\$80) (A) = (\$80) then anch to LP1
* * LPA	A is pro Memory HC05 cod CMP BEQ BRA HC08 cod	edefined at \$40 location \$80 con de \$80 LP1 LPA de	ntains * ;3,2 ;3,2 ; ; ;	\$40 (A) if bra go	) (A) = (\$80) then anch to LP1 to LPA, try again!
* * LPA	A is pro Memory HC05 coo CMP BEQ BRA	edefined at \$40 location \$80 con de \$80 LP1 LPA	ntains * ;3,2 ;3,2 ; ; ; ;	\$40 (A) if bra go if	<pre>(\$80) (A) = (\$80) then anch to LP1 to LPA, try again! (A)-(\$80)=0,</pre>
* * LPA	A is provide the second	edefined at \$40 location \$80 con de \$80 LP1 LPA de \$80,LPB	ntains * ;3,2 ;3,2 ; ; ; ; ; ; ; ; ; ; ; ; ;	\$40 (A) if bra go if the	<pre>(A) = (\$80) (A) = (\$80) then anch to LP1 to LPA, try again! (A)-(\$80)=0, en branch to LPB</pre>
* * LPA	A is pro Memory HC05 cod CMP BEQ BRA HC08 cod	edefined at \$40 location \$80 con de \$80 LP1 LPA de	ntains * ;3,2 ;3,2 ; ; ; ; ; ; ; ; ; ; ; ; ;	\$40 (A) if bra go if the	<pre>(\$80) (A) = (\$80) then anch to LP1 to LPA, try again! (A)-(\$80)=0,</pre>
* * LPA * LP1	A is provide the second	edefined at \$40 location \$80 con de \$80 LP1 LPA de \$80,LPB LP1	ntains * ;3,2 ;3,2 ; ; ; ; ; ; ; ; ; ; ; ; ; ; ; ; ; ; ;	\$40 (A) if bra go if the go	<ul> <li>(A) = (\$80)</li> <li>(A) = (\$80) then</li> <li>(A) to LP1</li> <li>(A) - (\$80) = 0,</li> <li>(A) - (\$80) = 0,</li> <li>(A) branch to LPB</li> <li>(A) LP1</li> </ul>
* * LPA * LP1	A is provide the second	edefined at \$40 location \$80 con de \$80 LP1 LPA de \$80,LPB LP1 PU05 cycles, by	ntains * ;3,2 ;3,2 ; ; ; ; ; ; ; ; ; ; ; ; ; ; ; ; ; ; ;	\$40 (A) if bra go if the go	<pre>(A) = (\$80) (A) = (\$80) then anch to LP1 to LPA, try again! (A)-(\$80)=0, en branch to LPB</pre>
*	A is provide the second	edefined at \$40 location \$80 con de \$80 LP1 LPA de \$80,LPB LP1	ntains * ;3,2 ;3,2 ; ; ; ; ; ; ; ; ; ; ; ; ; ; ; ; ; ; ;	\$40 (A) if bra go if the go	<pre>) (A) = (\$80) then anch to LP1 to LPA, try again! (A)-(\$80)=0, en branch to LPB to LP1 = 6,4</pre>
* * LPA * LP1	A is pro Memory HC05 cod CMP BEQ BRA HC08 cod CBEQ BRA Total C Total C	edefined at \$40 location \$80 con de \$80 LP1 LPA de \$80,LPB LP1 PU05 cycles, by	<pre>htains  *  ;3,2  ;3,2  ;  ;  ;  ;  ;5,3  ;  tes tes</pre>	\$40 (A) if bra go if the go	<pre>(A) = (\$80) (A) = (\$80) then anch to LP1 to LPA, try again! (A)-(\$80)=0, en branch to LPB to LP1 = 6,4 = 5,3</pre>
* * LPA * LP1 * *	A is pro- Memory HC05 cod CMP BEQ BRA HC08 cod CBEQ BRA Total C: Total C: CBEQA -	edefined at \$40 location \$80 con de \$80 LP1 LPA de \$80,LPB LP1 PU05 cycles, by PU08 cycles, by	<pre>htains  *  ;3,2  ;3,2  ;  ;  ;  ;  ;5,3  ;  tes tes</pre>	\$40 (A) if bra go if the go	<pre>(A) = (\$80) (A) = (\$80) then anch to LP1 to LPA, try again! (A)-(\$80)=0, en branch to LPB to LP1 = 6,4 = 5,3</pre>
* * LPA * LP1 * *	A is provide the second	edefined at \$40 location \$80 con de \$80 LP1 LPA de \$80,LPB LP1 PU05 cycles, by PU08 cycles, by compare and bra edefined at \$50	<pre>htains  *  ;3,2  ;3,2  ;  ;  ;  ;  ;5,3  ;  tes tes</pre>	\$40 (A) if bra go if the go	<pre>(A) = (\$80) (A) = (\$80) then anch to LP1 to LPA, try again! (A)-(\$80)=0, en branch to LPB to LP1 = 6,4 = 5,3</pre>
* * LPA * LP1 * * *	A is pro- Memory A HC05 coo CMP BEQ BRA HC08 coo CBEQ BRA Total C Total C CBEQA - A is pro- HC05 coo	edefined at \$40 location \$80 con de \$80 LP1 LPA de \$80,LPB LP1 PU05 cycles, by PU08 cycles, by compare and bra edefined at \$50 de	<pre>htains  *  ;3,2  ;3,2  ;  ;  ;  ;  ;  ;5,3  ;  tes tes anch i:  *</pre>	\$40 (A) if bra go if the go	<pre>) (A) = (\$80) then anch to LP1 to LPA, try again! (A)-(\$80)=0, en branch to LPB to LP1 = 6,4 = 5,3 gual, immediate</pre>
* * LPA * LP1	A is provide the second	edefined at \$40 location \$80 con de \$80 LP1 LPA de \$80,LPB LP1 PU05 cycles, by PU08 cycles, by compare and bra edefined at \$50	<pre>htains     *     ;3,2     ;3,2     ;3,2     ;     ;     ;5,3     ; tes tes anch i:     *     ;2,2</pre>	\$40 (A) if bra go if the go f eq (A)	<pre>(A) = (\$80) (A) = (\$80) then anch to LP1 to LPA, try again! (A)-(\$80)=0, en branch to LPB to LP1 = 6,4 = 5,3</pre>

AN1218 Rev. 2

* LP2	HC08 code * CBEQA #\$50,LPC ;4,3 if #\$50 = (A), then LPC BRA LP2 ; go to LP2
*	Total CPU05 cycles, bytes = 5,4 Total CPU08 cycles, bytes = 4,3
* * * * * *	CBEQX - compare and branch if equal, index Index register X is predefined at \$60
* LPC	HC05 code * CPX #\$60 ;2,2 X - \$60 BEQ LP3 ;3,2 if X = \$60, then LP3 BRA LPC ; go to LPC
* LP3	HC08 code * CBEQX #\$60,LPD ;4,3 if X = \$60, then LPD BRA LP3 ; go to LP3
*	Total CPU05 cycles, bytes = 5,4 Total CPU08 cycles, bytes = 4,3
* * * * *	DBNZ - decrement and branch if not zero
* * LPD	HC05 code * Memory location \$A0 is predefined at \$08 NOP ; used here to represent any ; number of instructions DEC \$A0 ;5,2 decrement (\$A0) BNE LPD ;3,2 if (\$A0) not zero, then LPD
* * LP4	HC08 code * Memory location \$A0 is predefined at \$08 NOP ; used here to represent any ; number of instructions DBNZ \$A0,LP4 ;5,3 (\$A0) = (\$A0) - 1 ; if (\$A0) not zero, then LP4
*	Total CPU05 cycles, bytes = 8,4 Total CPU08 cycles, bytes = 5,3

* * * * *	DBNZA - decrement acca and branch if not zero
*	HC05 code * A is predefined at \$06
LPE	NOP ; used here to represent any ; number of instructions
	DECA       ;3,1 (A) = (A) - 1         BNE       LPE       ;3,2 if (A) not zero, then LPE
*	HC08 code *
*	A is predefined at \$06
LP5	NOP ; used here to represent any ; number of instructions
	DBNZA LP5 $i3, 2(A) = (A) - 1$
	; if (A) not zero, then LP5
*	Total CPU05 cycles, bytes = 6,3
*	Total CPU08 cycles, bytes = 3,2
* * * * *	DBNZX - decrement x and branch if not zero
*	HC05 code *
*	Index register X is predefined at \$04
LPF	NOP ; used here to represent any
	; number of instructions 2 + 1 + (x) = -(x) + 1
	DECX $;3,1(X) = (X) - 1$ BNE LPF $;3,2$ if $(X)$ not zero, then LPF
*	HC08 code *
*	Index register X is predefined at \$04
LP6	NOP ; used here to represent any ; number of instructions
	$; number of instructions$ $DBNZX LP6 \qquad ;3,2 (X) = (X) - 1$
	; if (X) not zero, then LP6
*	Total CPU05 cycles, bytes = 6,3 Total CPU08 cycles, bytes = 3,2
	Iotal Croos Cycles, Dytes = 3,2
DONE	NOP DDA DONE
	BRA DONE
* * * * *	Initialize the reset vector
	ORG \$FFFE
	DW LPA

# Appendix H — CPU05 and CPU08 Search Code

******	• • • • • • • • • • •	* * * * * * * * * * * * * * * * * * *	* * * * * * * * * * * * * * * * * * * *	
*******	* * * * * * * * *	* * * * * * * * * * * * * * * * * * *	* * * * * * * * * * * * * * * * * * * *	
*	Filo ·	BRANCHX.ASM		
*	Descript			
*	Deseript	This code shows an example of using branch		
*			search for a number in a	
*		-	de will search for \$FF in	
*			would signify that in a	
*			alues, an A/D reading	
*				
*	Notes:	was saturated. Comments to the	right of some instructions	
*		give numbers.		
*		-	s CPU05 cycle count	
*			s instruction byte count	
*			s CPU08 cycle count	
*			s instruction byte count	
*			the CPU08 Reference Manual	
*		for further de	tails on these instructions	
<del>k</del>		Code is written	for educational	
k		purposes only		
r				
TABLE	EQU	\$400	; starting address of the ; A/D table	
	ORG	\$50		
BL_LEN	RMB	1	; memory value containing ; the number of values in ; a the A/D table	
	ORG	\$200		
* * * * * *	TBL_LI There The valu reflec	EN is predefined fore the table is les given for the	ion) in a table of A/D values at \$08 for this example s defined from \$400 to \$407 e total # of cycles and bytes ount with no looping involved the cycle count would involve	

# **Application Note**

*	HC05 cc	de	*	
SRCH	LDX	TBL_LEN	;3,2	$X \leftarrow (TBL_LEN)$
LOOP3	LDA	TABLE-1,X	;5,3	$A \leftarrow (TABLE-1+X)$
	CMP	#\$FF		(A) - \$FF
	BEQ	NEXT		if Z=1, then goto NEXT
	-			this signifies that a
			;	
			;	found in the table
	DECX		;3,1	X ← X −1
	BNE	LOOP3		if Z=0, then goto LOOP3
			;	go look at another value
			;	in the table
*	Total #	CPU05 cycles	= 19	
*	Total #	bytes	= 12	
*	HC08 cc	ode	*	
NEXT	LDX	TBL_LEN	;3,2	$X \leftarrow (TBL_LEN)$
LOOP4	LDA	TABLE-1,X		$A \leftarrow (TABLE-1+X)$
	CBEQA	#\$FF,DONE	;4,3	(A) - \$FF
			;	if Z=1, then goto DONE
			;	
			;	saturation value has been
			;	found in the table
	DBNZX	LOOP4		$x \leftarrow x - 1$
			;	-, 5
			;	go look at another value
			;	in the table
*	Шања] Ш		= 14	
*	Total #	CPU08 cycles	= 14 = 10	
	IULAI #	byles	- 10	
DONE	NOP			
Done	BRA	DONE		
* * * * *	Initial	ize the reset ve	ector	
	ORG	\$FFFE		
	DW	SRCH		

AN1218 Rev. 2

### Appendix I — New CPU08 Signed Branch Instruction Examples

```
*
       File : SIGNBRA.ASM
*
       Description :
*
               Shows examples for new CPU08 signed branch
                instructions - BGE, BGT, BLE, BLT
               The examples demonstrate two's complement
*
               math with branching.
       Note : Please consult the CPU08 Reference Manual
                for further details on these instructions
               Code is written for educational
*
               purposes only
*
ORG
               $200
* * * * *
       BGE - branch if greater than or equal
       A is predefined at $FF
LP_BGE
       CMP
               #$FF
                              ; (A) - $FF, [ -1 - (-1) ]
                              ; if A >= $FF, then
       BGE
               LP_BGT
                              ; branch to LP BGT
               LP BGE
       BRA
                              ; go to LP_BGE
* * * * *
       BGT - branch if greater than
*
       A is predefined at $07
LP BGT
                              ; (A) - $FF, [ 7 - (-1) ]
       CMP
               #$FF
                              ; if A > $FF, then
               LP_BLE
       BGT
                              ; branch to LP_BLE
       BRA
               LP BGT
                              ; go to LP_BGT
* * * * *
       BLE - branch if less than or equal
       A is predefined at $FF
LP_BLE
       CMP
               #$FF
                              ; (A) - $FF, [ -1 - (-1) ]
                              ; if A <= $FF, then
       BLE
               LP_BLT
                              ; branch to LP_BLT
       BRA
               LP_BLE
                              ; go to LP_BLE
* * * * *
       BLT - branch if less than
*
       A is predefined at $FF
LP BLT
       CMP
               #$07
                              ; $FF - $07, [ -1 - (7) ]
                              ; if A < \$FF, then
       BLT
               DONE
                              ; branch to DONE
       BRA
               LP_BLT
                              ; go to LP_BLT
DONE
       NOP
       BRA
               DONE
* * * * *
       Initialize the reset vector
       ORG
               $FFFE
       DW
               LP BGE
```

# **Application Note**

Appendix J — Five Miscellaneous CPU08 Instructions Including BCD, Divide, and CCR Operations

*	File :	MTOOTNOD NOM		
*				
*	Description : Shows examples for 5 misc CPU08 instruction			
*				
*			CD, Divide, and CCR operatio	
*	NT 1 .		NSA, DIV, TAP, TPA	
*	Notes:	give numbers.	e right of some instructions	
*		CPU05 - 1st #	is CPU05 cycle count	
*		2nd #	is instruction byte count	
*		CPU08 - 1st #	is CPU08 cycle count	
*		2nd #	is instruction byte count	
*		Please consult	the CPU08 Reference Manual	
*		for further d	etails on these instructions	
*		Code is writte	n for educational	
*		purposes only		
*				
* * * * * * *	* * * * * * * * *	*****	* * * * * * * * * * * * * * * * * * * *	
	ORG	\$200		
* * * * *	-			
* * * * *	DAA – d	lecimal adjust a	ccumulator	
START	LDA	#\$26	; A $\leftarrow$ \$26, a BCD #	
DIARI	ADD	#\$20 #\$37	$; A \leftarrow $37 + (A)$	
	TIDD .	πφυ /	; $A = $5D$ , a hex #	
	DAA		(A) = 63 = (26 + 37)	
	DAA		; the hex #, 5D, has been	
			; adjusted to the BCD #, 6	
			, adjusted to the BCD #, o	
* * * * *	NSA – n	ibble swap accu	mulator	
*		redefined at \$37		
*		nished A will b		
	WIICH II			
*	HC05 co	de	*	
	TAX		;2,1 X $\leftarrow$ (A)	
	ROLX		;3,1 rotate left X	
	ROLA		;3,1 rotate left A	
	ROLX		;3,1 rotate left X	
	ROLA		;3,1 rotate left A	
	ROLX		;3,1 rotate left X	
	TOTIX			
	POT A		$: $   rotate left $\Lambda$	
	ROLA ROLX		;3,1 rotate left A ;3,1 rotate left X	

AN1218 Rev. 2

* HC08 code * NSA ;3,1 swap the nibbles of A Total CPU05 cycles, bytes * = 26,9 * Total CPU08 cycles, bytes = 3,1 * * * * * DIV - divide 16 bit by 8 bit * The immediate addressing mode is used to load the registers * to illustrate the components needed to execute * a DIV instruction. LDHX #\$0200 ; H ← \$02 LDX #\$80 ; X ← \$80 LDA #\$00 ;  $A \leftarrow \$00$ DIV ; H:A / X = A rem H ; Answer is \$04 rem 0 * * * * * TAP - transfer accumulator to ccr A is predefined at \$E2 * CCR = %0110,0000TAP ; CCR  $\leftarrow$  (A) ; CCR = %1110,0010 * * * * * TPA - transfer ccr to accumulator A is predefined at \$00 * * CCR = %1110,0010 TPA ;  $A \leftarrow (CCR)$ ; A = \$E2DONE NOP BRA DONE * * * * * Initialize the reset vector ORG \$FFFE DW START

### Appendix K — CPU08 Averaging Code

* * * * * * * * * * * * * * * File : AVERAGE.ASM * Description : This code demonstrates an average routine showing the use of the CPU08's DIV inst. 8-bit values are read from a table in memory and the average of those numbers * is computed. Notes: Please consult the CPU08 Reference Manual for further details on these instructions Code is written for educational purposes only TBL_STR EQU \$400 ; starting address of ; table in memory ORG \$50 LENGTH RMB 1 ; length of table тот н RMB 1 ; high byte of total TOTL ; low byte of total RMB 1 \$200 ORG * The length of this table is predefined as 3 for * this example * The values in the table start at \$401 * \$401 = 50* \$402 = 60\$403 = 70* ; clear TOT H in mem START CLR ТОТ Н CLR TOT L ; clear TOT_L in mem LDX LENGTH ;  $X \leftarrow$  length of table NEXT LDA TBL_STR,X ; A  $\leftarrow$  (X+TBL_STR) ADD TOT_L ;  $A \leftarrow (A) + (TOT_L)$ STA TOT_L ; TOT_L  $\leftarrow$  (A) ; if carry bit is set, BCS CS ; branch to CS ; branch to next table entry BRA NEXT2 CS INC ; inc the high byte of total TOT_H ; dec X, if X not 0, then NEXT2 DBNZX NEXT branch to next table entry ; LDHX TOT_H ;  $H \leftarrow high byte of dividend$ TXA ;  $A \leftarrow low byte of dividend$ ;  $X \leftarrow \text{load divisor}$ ; H:A / X LDX LENGTH DIV ; answer in A with H rem Answer can be found in A, remainder in H Answer is equal to \$60 with no remainder DONE NOP BRA DONE * * * * * Initialize the reset vector ORG \$FFFE START DW

### Appendix L — CPU08 BCD Example Code

```
File : BCD.ASM
       Description :
*
               This code demonstrates a BCD routine to
               be used on the CPU08.
               Two 16-bit BCD numbers are added together
               and the result is 16-bit BCD number
               BCD1 + BCD2 = BCDT
       Notes: Please consult the CPU08 Reference Manual
               for further details on these instructions
               Code is written for educational
               purposes only
$50
       ORG
BCD1 H RMB
                              ; high byte of bcd #1
               1
BCD1 L
                              ; low byte of bcd #1
               1
       RMB
BCD2 H RMB
                             ; high byte of bcd #2
               1
BCD2 L RMB
              1
                             ; low byte of bcd #2
BCDT H RMB
                             ; high byte of bcd total
              1
BCDT L RMB
               1
                              ; low byte of bcd total
               $200
       ORG
*
       Predefine values for the example
*
       BCD1 = 150, BCD1_H = 01 & BCD1_L = 50
       BCD2 = 250,
                    BCD2_H = 02 \& BCD2_L = 50
*
       First, add the low bytes of the 16-bit BCD #s
*
START
       LDA
               BCD1 L
                         ; A \leftarrow (bcd \#1 low byte)
       ADD
               BCD2 L
                              ; A \leftarrow (A) + (bcd \#2)
                                low byte)
                              ;
       DAA
                              ; decimal adjust accumulator
       STA
              BCDT_L
                             ; store away result to total low
       Second, add the high bytes of the 16-bit BCD #s
*
*
       Add the carry bit from the previous addition
               BCD1 H
                              ; A \leftarrow (bcd \#1 high byte)
       LDA
                              ; A \leftarrow (A) + (bcd \#2)
       ADC
               BCD2 H
                              ; high byte)+C
                              ; decimal adjust accumulator
       DAA
       STA
              BCDT_H
                             ; store away result to total high
       Answer is in BCDT_H and BCDT_L
       BCDT H = 04
*
       BCDT L = 00
DONE
       NOP
               DONE
       BRA
* * * * *
       Initialize the reset vector
       ORG
              ŚFFFE
       DW
               START
```

Motorola reserves the right to make changes without further notice to any products herein. Motorola makes no warranty, representation or guarantee regarding the suitability of its products for any particular purpose, nor does Motorola assume any liability arising out of the application or use of any product or circuit, and specifically disclaims any and all liability, including without limitation consequential or incidental damages. "Typical" parameters which may be provided in Motorola data sheets and/or specifications can and do vary in different applications and actual performance may vary over time. All operating parameters, including "Typicals" must be validated for each customer application by customer's technical experts. Motorola does not convey any license under its patent rights of others. Motorola products are not designed, intended, or authorized for use as components in systems intended for surgical implant into the body, or other applications intended to support or sustain life, or for any other application in which the failure of the Motorola product could create a situation where personal injury or death may occur. Should Buyer purchase or use Motorola products for any such unintended or unauthorized application, Buyer shall indemnify and hold Motorola and its officers, employees, subsidiaries, affiliates, and distributors harmless against all claims, costs, damages, and expenses, and reasonable attorney fees arising out of, directly or indirectly, any claim of personal injury or death masciated with such unintended or unauthorized use, even if such claim alleges that Motorola was negligent regarding the design or manufacture of the part. Motorola and (**A**) are registered trademarks of Motorola, Inc. Motorola, Inc. is an Equal Opportunity/Affirmative Action Employer.

#### How to reach us:

USA/EUROPE/Locations Not Listed: Motorola Literature Distribution, P.O. Box 5405, Denver, Colorado 80217, 1-800-441-2447 or 1-303-675-2140. Customer Focus Center, 1-800-521-6274

JAPAN: Nippon Motorola Ltd.: SPD, Strategic Planning Office, 141, 4-32-1 Nishi-Gotanda, Shinagawa-ku, Tokyo, Japan. 03-5487-8488 ASIA/PACIFIC: Motorola Semiconductors H.K. Ltd., 8B Tai Ping Industrial Park, 51 Ting Kok Road, Tai Po, N.T., Hong Kong. 852-26629298 Mfax™, Motorola Fax Back System: RMFAX0@email.sps.mot.com; http://sps.motorola.com/mfax/;

TOUCHTONE, 1-602-244-6609; US and Canada ONLY, 1-800-774-1848

HOME PAGE: http://motorola.com/sps/



Mfax is a trademark of Motorola, Inc.

© Motorola, Inc., 1993